NAG Fortran Library Routine Document

F08MEF (SBDSQR/DBDSQR)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

Warning. The specification of the parameter WORK changed at Mark 20: the length of WORK needs to be increased.

1 Purpose

F08MEF (SBDSQR/DBDSQR) computes the singular value decomposition of a real upper or lower bidiagonal matrix, or of a real general matrix which has been reduced to bidiagonal form.

2 Specification

```
      SUBROUTINE F08MEF(UPLO, N, NCVT, NRU, NCC, D, E, VT, LDVT, U, LDU, C,
     1                  LDC, WORK, INFO)
      ENTRY      sbdsqr (UPLO, N, NCVT, NRU, NCC, D, E, VT, LDVT, U, LDU, C,
     1                  LDC, WORK, INFO)
      INTEGER           N, NCVT, NRU, NCC, LDVT, LDU, LDC, INFO
      real              D(*), E(*), VT(LDVT,*), U(LDU,*), C(LDC,*), WORK(*)
      CHARACTER*1       UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes the singular values, and optionally, the left or right singular vectors of a real upper or lower bidiagonal matrix B. In other words, it can compute the singular value decomposition (SVD) of B as

$$B = U\Sigma V^T.$$

Here Σ is a diagonal matrix with real diagonal elements σ_i (the singular values of B), such that

$$\sigma_1 \geq \sigma_2 \geq \ldots \geq \sigma_n \geq 0;$$

U is an orthogonal matrix whose columns are the left singular vectors u_i; V is an orthogonal matrix whose rows are the right singular vectors v_i. Thus

$$Bu_i = \sigma_i v_i \quad \text{and} \quad B^T v_i = \sigma_i u_i \quad \text{for} \quad i = 1, 2, \ldots, n.$$

To compute U and/or V^T, the arrays U and/or VT must be initialised to the unit matrix before F08MEF (SBDSQR/DBDSQR) is called.

The routine may also be used to compute the SVD of a real general matrix A which has been reduced to bidiagonal form by an orthogonal transformation: $A = QBP^T$. If A is m by n with $m \geq n$, then Q is m by n and P^T is n by n; if A is n by p with $n < p$, then Q is n by n and P^T is n by p. In this case, the matrices Q and/or P^T must be formed explicitly by F08KFF (SORGBR/DORGBR) and passed to F08MEF (SBDSQR/DBDSQR) in the arrays U and/or VT respectively.

F08MEF (SBDSQR/DBDSQR) also has the capability of forming $U^T C$, where C is an arbitrary real matrix; this is needed when using the SVD to solve linear least-squares problems.

F08MEF (SBDSQR/DBDSQR) uses two different algorithms. If any singular vectors are required (i.e., if NCVT > 0 or NRU > 0 or NCC > 0), the bidiagonal QR algorithm is used, switching between zero-shift and implicitly shifted forms to preserve the accuracy of small singular values, and switching between QR and QL variants in order to handle graded matrices effectively (see Demmel and Kahan (1990)). If only singular values are required (that is, if NCVT $=$ NRU $=$ NCC $= 0$), they are computed by the differential qd algorithm (see Fernando and Parlett (1994)), which is faster and can achieve even greater accuracy.

The singular vectors are normalized so that $\|u_i\| = \|v_i\| = 1$, but are determined only to within a factor ± 1.

4 References

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Fernando K V and Parlett B N (1994) Accurate singular values and differential qd algorithms *Numer. Math.* **67** 191–229

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether B is an upper or lower bidiagonal matrix as follows:

if UPLO = 'U', B is an upper bidiagonal matrix;

if UPLO = 'L', B is a lower bidiagonal matrix.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix B.

Constraint: N $\geq$ 0.

3: NCVT – INTEGER *Input*

On entry: $ncvt$, the number of columns of the matrix V^T of right singular vectors. Set NCVT = 0 if no right singular vectors are required.

Constraint: NCVT $\geq$ 0.

4: NRU – INTEGER *Input*

On entry: nru, the number of rows of the matrix U of left singular vectors. Set NRU = 0 if no left singular vectors are required.

Constraint: NRU $\geq$ 0.

5: NCC – INTEGER *Input*

On entry: ncc, the number of columns of the matrix C. Set NCC = 0 if no matrix C is supplied.

Constraint: NCC $\geq$ 0.

6: D($*$) – ***real*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: the diagonal elements of the bidiagonal matrix B.

On exit: the singular values in decreasing order of magnitude, unless INFO > 0 (in which case see Section 6).

7: E($*$) – ***real*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: the off-diagonal elements of the bidiagonal matrix B.

On exit: the array is overwritten, but if INFO > 0 see Section 6.

8: VT(LDVT,*) – ***real*** array *Input/Output*

Note: the second dimension of the array VT must be at least $\max(1, \mathrm{NCVT})$.

On entry: if NCVT > 0, VT must contain an n by $ncvt$ matrix. If the right singular vectors of B are required, $ncvt = n$ and VT must contain the unit matrix; if the right singular vectors of A are required, VT must contain the orthogonal matrix P^T returned by F08KFF (SORGBR/DORGBR) with VECT = 'P'.

On exit: the n by $ncvt$ matrix V^T or $V^T P^T$ of right singular vectors, stored by rows.

VT is not referenced if NCVT $= 0$.

9: LDVT – INTEGER *Input*

On entry: the first dimension of the array VT as declared in the (sub)program from which F08MEF (SBDSQR/DBDSQR) is called.

Constraints:

LDVT $\geq \max(1, \mathrm{N})$ if NCVT > 0,
LDVT ≥ 1 otherwise.

10: U(LDU,*) – ***real*** array *Input/Output*

Note: the second dimension of the array U must be at least $\max(1, \mathrm{N})$.

On entry: if NRU > 0, U must contain an nru by n matrix. If the left singular vectors of B are required, $nru = n$ and U must contain the unit matrix; if the left singular vectors of A are required, U must contain the orthogonal matrix Q returned by F08KFF (SORGBR/DORGBR) with VECT = 'Q'.

On exit: the nru by n matrix U or QU of left singular vectors, stored by columns.

U is not referenced if NRU $= 0$.

11: LDU – INTEGER *Input*

On entry: the first dimension of the array U as declared in the (sub)program from which F08MEF (SBDSQR/DBDSQR) is called.

Constraint: LDU $\geq \max(1, \mathrm{NRU})$.

12: C(LDC,*) – ***real*** array *Input/Output*

Note: the second dimension of the array C must be at least $\max(1, \mathrm{NCC})$.

On entry: the n by ncc matrix C if NCC > 0.

On exit: C is overwritten by the matrix $U^T C$.

C is not referenced if NCC $= 0$.

13: LDC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which F08MEF (SBDSQR/DBDSQR) is called.

Constraints:

LDC $\geq \max(1, \mathrm{N})$ if NCC > 0,
LDC ≥ 1 otherwise.

14: WORK($*$) – ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 4 * \mathrm{N})$.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm failed to converge and INFO specifies how many off-diagonals did not converge. In this case, D and E contain on exit the diagonal and off-diagonal elements, respectively, of a bidiagonal matrix orthogonally equivalent to B.

7 Accuracy

Each singular value and singular vector is computed to high relative accuracy. However, the reduction to bidiagonal form (prior to calling the routine) may exclude the possibility of obtaining high relative accuracy in the small singular values of the original matrix if its singular values vary widely in magnitude.

If σ_i is an exact singular value of B and $\tilde{\sigma}_i$ is the corresponding computed value, then

$$|\tilde{\sigma}_i - \sigma_i| \leq p(m,n)\epsilon\sigma_i$$

where $p(m,n)$ is a modestly increasing function of m and n, and ϵ is the ***machine precision***. If only singular values are computed, they are computed more accurately (i.e., the function $p(m,n)$ is smaller), than when some singular vectors are also computed.

If u_i is the corresponding exact left singular vector of B, and $\tilde{u}_i$ is the corresponding computed left singular vector, then the angle $\theta(\tilde{u}_i, u_i)$ between them is bounded as follows:

$$\theta(\tilde{u}_i, u_i) \leq \frac{p(m,n)\epsilon}{relgap_i}$$

where $relgap_i$ is the relative gap between σ_i and the other singular values, defined by

$$relgap_i = \min_{i \neq j} \frac{|\sigma_i - \sigma_j|}{(\sigma_i + \sigma_j)}.$$

A similar error bound holds for the right singular vectors.

8 Further Comments

The total number of floating-point operations is roughly proportional to n^2 if only the singular values are computed. About $6n^2 \times nru$ additional operations are required to compute the left singular vectors and about $6n^2 \times ncvt$ to compute the right singular vectors. The operations to compute the singular values must all be performed in scalar mode; the additional operations to compute the singular vectors can be vectorized and on some machines may be performed much faster.

The complex analogue of this routine is F08MSF (CBDSQR/ZBDSQR).

9 Example

To compute the singular value decomposition of the upper bidiagonal matrix B, where

$$B = \begin{pmatrix} 3.62 & 1.26 & 0.00 & 0.00 \\ 0.00 & -2.41 & -1.53 & 0.00 \\ 0.00 & 0.00 & 1.92 & 1.19 \\ 0.00 & 0.00 & 0.00 & -1.43 \end{pmatrix}.$$

See also the example for F08KFF (SORGBR/DORGBR), which illustrates the use of the routine to compute the singular value decomposition of a general matrix.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08MEF Example Program Text
*     Mark 20 Revised. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDVT, LDU, LDC
      PARAMETER        (NMAX=8,LDVT=NMAX,LDU=NMAX,LDC=1)
      real             ZERO, ONE
      PARAMETER        (ZERO=0.0e0,ONE=1.0e0)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             C(LDC,1), D(NMAX), E(NMAX-1), U(LDU,NMAX),
     +                 VT(LDVT,NMAX), WORK(4*NMAX)
*     .. External Subroutines ..
      EXTERNAL         sbdsqr, F06QHF, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08MEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read B from data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
         READ (NIN,*) UPLO
*
*        Initialise U and VT to be the unit matrix
*
         CALL F06QHF('General',N,N,ZERO,ONE,U,LDU)
*
         CALL F06QHF('General',N,N,ZERO,ONE,VT,LDVT)
*
*        Calculate the SVD of B
*
         CALL sbdsqr(UPLO,N,N,N,0,D,E,VT,LDVT,U,LDU,C,LDC,WORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.GT.0) THEN
            WRITE (NOUT,*) 'Failure to converge.'
         ELSE
*
*           Print singular values, left & right singular vectors
*
            WRITE (NOUT,*) 'Singular values'
            WRITE (NOUT,99999) (D(I),I=1,N)
            WRITE (NOUT,*)
```

```
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,N,VT,LDVT,
     +                  'Right singular vectors, by row',IFAIL)
*
            WRITE (NOUT,*)
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,N,U,LDU,
     +                  'Left singular vectors, by column',IFAIL)
*
         END IF
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
      END
```

9.2 Program Data

```
F08MEF Example Program Data
  4                            :Value of N
  3.62  -2.41   1.92  -1.43
  1.26  -1.53   1.19           :End of matrix B
  'U'                          :Value of UPLO
```

9.3 Program Results

```
 F08MEF Example Program Results

 Singular values
     4.0001  3.0006  1.9960  0.9998

 Right singular vectors, by row
          1       2       3       4
 1   0.8261  0.5246  0.2024  0.0369
 2   0.4512 -0.4056 -0.7350 -0.3030
 3   0.2823 -0.5644  0.1731  0.7561
 4   0.1852 -0.4916  0.6236 -0.5789

 Left singular vectors, by column
          1       2       3       4
 1   0.9129  0.3740  0.1556  0.0512
 2  -0.3935  0.7005  0.5489  0.2307
 3   0.1081 -0.5904  0.6173  0.5086
 4  -0.0132  0.1444 -0.5417  0.8280
```